
LE FORMAT GRAPHIQUE SVG ET SAS 

 
Introduit avec la version 9.2 de SAS, et largement 
amélioré dans la version 9.3, le format SVG (Scalable 

Vector Graphics) est le standard graphique du web défini 
par le W3C, particulièrement adapté à la publication de 
vos rapports graphiques sur votre intranet ou votre site 
internet.  

Le format SVG 

Le format SVG (Scalable Vector Graphics) est un format XML, qui décrit une image vectorielle en 
deux dimensions. Ses principaux avantages sont donc : 

- La possibilité de redimensionner l’image, sans perte de qualité, 

- La taille des fichiers : ils sont plus petits que leurs équivalents au format GIF ou PNG, et sont 

indépendants de la taille d’affichage. 

Ce format est supporté par un nombre grandissant de navigateurs internet :  

- Mozilla Firefox, depuis la version 1.5 

- Apple Safari, depuis la version 3 

- Google Chrome 

- Microsoft Internet Explorer, depuis la version 9 

Note : L’ODS html de SAS produit un fichier au standard HTLM 4.0, et dans ce mode, Internet 

Explorer 9 ne restitue pas correctement les graphiques SVG complexes (animations…). Il faut alors 
utiliser l’ODS tagsets.xhtml à la place de l’ODS html. 

Pour les versions précédentes d’Internet Explorer, il est possible d’utiliser le plug-in Adobe SVG 
Viewer, téléchargeable à cette adresse : http://www.adobe.com/svg/viewer/install/ 
Notez toutefois qu’Adobe n’assure plus de support pour ce composant. 

Générer des graphiques SVG avec SAS 

Il y a quatre sorties graphiques (ou « devices ») SVG disponibles avec SAS : 

- SVG produit des graphiques à la norme SVG 1.1 

- SVGT produit des graphiques à la norme SVG 1.1, avec un arrière-plan transparent 

- SVGZ produit un fichier SVG compressé, ce qui peut être intéressant quand les graphiques sont 

conséquents, mais qui n’est pas supporté par tous les navigateurs. 

- SVGVIEW est utilisé avec la sortie ODS LISTING, et ajoute des contrôles de navigation entre les 

graphes, quand plusieurs graphes sont générés par une même procédure. 

 Caractéristiques : 

Catégories : SAS/GRAPH® 
OS : Windows, Unix, z/OS 
Version : SAS® 9.3 

Vérifié en Septembre 2012 

http://www.sas.com/fr_fr/support.html
http://fr.wikipedia.org/wiki/W3C
http://www.adobe.com/svg/viewer/install/

Exemple simple 

L’option graphique device=svg dirige la sortie graphique standard vers un fichier SVG externe,  
nommé par défaut du nom de la procédure graphique (ici plot.svg). 
 

goptions reset=all device=svg border; 

ods html file="plot.html"; 

title "Relation Poids-Taille"; 

footnote; 

 

proc gplot data=sashelp.class; 

 symbol1 interpol=rcclm95 value=circle cv=darkred  

     ci=black co=blue width=2; 

 axis1 label=("Poids") order=(45 to 155 by 10); 

 axis2 label=("Taille") order=(48 to 78 by 6); 

 plot height*weight / haxis=axis1 vaxis=axis2 hminor=1; 

 run; 

quit; 

 

ods html close; 

Voir le fichier GraphiquesSVG.gplot.svg dans ce zip.  
 

Les nouveautés 9.3 

Directives HTML 

Il est maintenant possible d’ajouter des directives HTML directement dans le graphique, comme 
c’était déjà le cas pour les sorties graphiques ActiveX ou Java. 

On peut par exemple définir une info-bulle, un lien hypertexte, ou une interaction avec la souris. 
 

Dans cet exemple, nous créons d’abord une variable myhtmlvar, qui contiendra les directives HTML : 

http://www.sas.com/fr_fr/support.html
http://www.sas.com/offices/europe/france/services/support/download/US2012_Q3_GraphiquesSVG.zip

- alt=  pour une info-bulle. 

- onmouseover=  pour une action au passage de la souris. 

- href=  pour un lien hypertexte. 

data cars; 

 set sashelp.cars; 

 length  myhtmlvar $400; 

 myhtmlvar='alt='||quote('Constructeur: '||trim(make)||' '||'0D'x|| 

    'Type: '||trim(type))||' '|| 

    'onmouseover=changeOpacity(.5) '|| 

    'href="http://www.'||trim(left(lowcase(make)))|| 

    '.com/'||trim(left(lowcase(type)))||'"'; 

run; 

 

Puis nous utilisons l’option html=  pour intégrer ces directives au graphique SVG. 
 

goptions reset=all device=svg; 

ods listing close; 

ods html body="gchart_svg.html"; 

title "Nombre de modèles par type"; 

axis1 label=("Constructeur"); 

axis2 label=("Nb de models"); 

proc gchart data=cars; 

 hbar make / description="Survolez une barre..."  

   subgroup=type nolegend nostats maxis=axis1 raxis=axis2 

   html=myhtmlvar;  

 run; 

quit; 

ods html close; 

ods listing; 

 

http://www.sas.com/fr_fr/support.html

Voir le fichier GraphiquesSVG.gchart.svg dans ce zip.  

 

SVG animé 

La génération de graphiques SVG animés se fait grâce à la nouvelle imprimante virtuelle ODS 
SVGAMIN. 
Dans l’exemple ci-dessous, nous allons générer 4 graphiques, que nous allons enregistrer dans un 
seul et même fichier de sortie. 

L’option graphique delay=  défini le temps d’affichage entre chaque graphique, en centièmes de 
seconde. 
 

ods listing close; 

goptions reset=all delay=200;  

ods printer file='animchart.svg' style=harvest printer=svganim; 

 

/* On supprime les graphiques existant */ 

proc greplay igout=work.gseg nofs; 

 delete _all_; 

run; quit; 

 

/* Préparation des données */ 

data work.qsales; 

 set sashelp.prdsale(where=(country="CANADA" or country="GERMANY") 

     keep=Actual Country Product Quarter Year); 

run; 

proc sort data=work.qsales; 

 by quarter; 

run; 

 

/* Procédure graphique */ 

title1 "1993 Sales"; 

title2 "Quarter #BYVAL(quarter)" ; 

axis1 order=(0 to 30000 by 5000); 

proc gchart data=work.qsales(where=(year=1993)); 

 vbar3d country / sumvar=actual subgroup=product sum raxis=axis1 

       shape=hexagon; 

  where product in ("BED" "TABLE" "CHAIR"); 

  by quarter; 

 run; 

quit; 

ods printer close; 

 
Par défaut, l’imprimante SVGAMIN ajoute au graphique des boutons de contrôle pour lancer et arrêter 
l’animation, revenir à l’image initiale, et masquer les contrôles. 

http://www.sas.com/fr_fr/support.html
http://www.sas.com/offices/europe/france/services/support/download/US2012_Q3_GraphiquesSVG.zip

Voir le fichier GraphiquesSVG.animchart.svg dans ce zip.  

